Maple Lecture 13. Substitution, Expansion and Factorization

Symbolic calculations often result in very long expressions which can be hard to interpret. Substitution is one of the major tools to simplify expressions. We will see how to apply the subs command to only one operator and learn the distinction between algebraic and formal substitution. Geometrically, the factorization of polynomials corresponds to finding the roots. The result of this operation depends heavily on the number field we work in. The same holds for the expansion of polynomial, an operation which is in many ways opposite to factorization.

This lecture note corresponds to [1, Sections 6.3, 7.1 and 7.2].

13.1 Formal and Algebraic Substitution

With the **subs** command we can control the simplification of expressions, e.g.:

```
[> p := (x+y)^2 + 1/(x+y)^2;
[> normal(p);
```

A better representation of p is to have $(x+y)^4+1$ in the numerator. We can achieve this by replacing $x+y$ by some variable z, normalize the substituted expression and the replace z by $x+y$ after the normalization.

```
[> pz := subs(x+y=z,p);
```

Observe the syntax of the command: in the equality we substitute left hand side by right hand side anywhere the left hand side occurs in the second argument of p.

```
[> npz := normal(pz);
[> q := subs(z=x+y,npz);
```

With **subsop** we restrict the substitution to one operand. Suppose we wish to replace $x+y$ by $x-y$ in the denominator only. With **op([2,1],q);** we verify that the denominator is the first element of second operand sequence of q, denoted by [2,1] (reading from left to right). Replacing $x+y$ by $x-y$ goes as follows:

```
[> subsop([2,1]=x-y,q);
```

In expressions with several variables, the order of substitution can matter a lot. We distinguish sequential and simultaneous substitution. Consider for example the permutation of variables.

```
[> p := a + 2*b + 3*c;
```

Suppose we wished to permute the variables in a cyclic way, **a** becomes **b**, **b** becomes **c** and **c** becomes **a**:

```
[> subs(a=b,b=c,c=a,p);
```

The above substitute executes the substitutions one after the other, which is equivalent to

```
[> subs(c=a,subs(b=c,subs(a=b,p)));
```

But in a cyclic permutation, we want the substitutions done simultaneously, all at once:

```
[> subs({a=b,b=c,c=a},p);
```

From your earlier algebra courses, you may remember that substitution is one way to solve linear equations and to simplify expressions.

Suppose we new that `a+b = c`, then we could simplify the expression for p above. We can try with subs:

```
[> subs(a+b=c,p);
```

This does not work, since subs does not recognize the algebraic structure. Instead we must use the algebraic substitution command:

```
[> algsubs(a+b=c,p);
```

The subs is a syntactical substitution, it only substitutes when there is an exact match for a+b, ignoring the mathematical structures.

13.2 Expansion of Polynomials and Algebraic Numbers

We have already seen that Maple does not expand automatically. To expand we have the expand and Expand commands. For example, if we multiply $a + b + c$ with some polynomial:

```
[> p := (a+b+c)*(x^3 +9*x + 8);
[> expand(p);
```

But suppose we wished to expand only partially, leaving the polynomial intact:

```
[> op(2,p);
[> q := subsop(2=z,p);
[> eq := expand(q);
[> ep := subs(z=op(2,p),eq);
```

For algebraic numbers, we use the Expand command, mainly in conjunction with modulo, when working over finite fields. The experiment below, shows what needs to be done when extending the rationals with a root of an irreducible polynomial:

```
[> p2 := op(2,p);
[> irreduc(p2);
[> alias(alpha=RootOf(p2));
```

Suppose we now calculate in $\mathbb{Q}(\alpha) = \{ \ a + b\alpha + c\alpha^2 \ | \ a, b, c \in \mathbb{Q} \ \}$.

```
[> ap := (alpha+1)^7;
```

Usually we would not wish to expand an expression as the above, but it simplifies because alpha is a root of a cubic polynomial:

```
[> expand(ap);
[> evala(%);
```

In this case, we can also simply try **simplify(ap)**.

13.3 Factorization: Exact, Symbolic, and Numeric

We can factor polynomials in three ways:

1) Exact: as in $x^2 - 1 = (x - 1)(x + 1)$, only possible when all roots are rational numbers.

2) Symbolically: We add formal roots, extending the field of rational numbers.

3) Numerically: A polynomial of degree d has d complex roots (counted with multiplicity).

With the **convert** and **evalf** we transform from symbolic respectively to exact and to numeric factorization. Below are some examples. Note the importance of the number field.

```
[> p := x^2 + 2;
[> factor(p);
```

This polynomial does not factor over the rational numbers, but we can extend the rational numbers with the square root of two and the imaginary unit :

```
[> factor(p,{sqrt(2),I});
```

Now we knew what we needed to add as extensions to the factor command. In most cases we will not have this information, therefore, Maple offers the command split as part of the polytools package:

```
[> fp := polytools[split](p,x);
```

At first it seems as if Maple has not done much, but you can convert to a nicer representation:

```
[> rfp := convert(fp,radical);
```

To see the imaginary unit appearing, we apply evalc (evaluate complex). Simply applying evalc to rfp would expand the polynomial, something we do not want. Instead we apply evalc to all the operands of the expression rfp, for which the command map is appropriate, e.g.: **map(evalc,rfp)**.

13.4 Assignments

1. Give the Maple commands to transform

$$(x+y)^2 + \frac{1}{x+y}$$

into

$$\frac{(x+y)^3+1}{x+y}$$

and vice versa.

2. Give the Maple commands to transform

$$x^2 + 2x + 1 + \frac{1}{x^2+2x+1}$$

into

$$\frac{(x+1)^4+1}{(x+1)^2}$$

and vice versa.

3. Give the Maple commands to transform

$$x^3 - xy^2 - yx^2 + y^3 + x^2 - y^2$$

into

$$(x^2-y^2)(x-y+1).$$

4. Give the Maple commands to transform

$$(x+z^2+1)(y-z^2-1)$$

into

$$xy - x(z^2+1) + (z^2+1)y - (z^2+1)^2.$$

5. Consider the polynomial $p = 92 - 93x - 8x^2$.

   Give the Maple command(s) to write p as

$$-8\left(x+\frac{93}{16}-\frac{\sqrt{11593}}{16}\right)\left(x+\frac{93}{16}-\frac{\sqrt{11593}}{16}\right).$$

References

[1] A. Heck. *Introduction to Maple*. Springer-Verlag, third edition, 2003.